NVSS Source Catalog

December 10, 1997

1 Overview

The NRAO/VLA Sky Survey (NVSS) is a survey of the sky using the VLA radio telescope array at a wavelength of 20 cm (1.4 GHz) covering the declination range north of δ = −40^o (Condon et al. 1998). The survey produces images in Stokes I, Q and U (total intensity and linear polarization) with a resolution of 45”. This document describes the NVSS source catalog tables derived from these images as stored in a FITS binary table.

2 NVSS Catalog table

2.1 General information

This table contains source information derived from the NVSS images. The entries consist of single Gaussian model fits and there is no attempt to identify which entries are associated with a particular physical source. Each component is identified and fitted in the Stokes I image and then polarization information is derived for these components from the Stokes Q and U images. Individual model components do not have associated errors in the table but these are assumed derivable from the model fit parameters and the image RMS noises in Stokes I and Q+U. The values in the table are those fitted to the image and the size will reflect the convolution with the restoring beam (BM MAJOR, BM MINOR, BM PA). In order to derive the actual size of the component the fitted Gaussian must be deconvolved from the restoring Gaussian.

This catalog is stored in FITS binary table format (Cotton et al. 1995. The header of the table may contain an index to speed access of an RA sorted table. Header keyword INDEXnn gives the record number of the first record whose Right Ascension exceeds nn hours.

2.2 Table Structure

The header keywords give general information about the table and possibly an index to speed access.

  Keyword type Description
--------- ---- ---------------------------------------------------
  REVISION I Revision number of the VL table format
  BM_MAJOR R Major axis FWHM of restoring beam in degrees.
  BM_MINOR R Minor axis FWHM of restoring beam in degrees.
  BM_PA R Position angle of the major axis of the restoring

3 FITTING CONSTRAINTS 2

              beam (from north through E) in degrees.
SORTORD I Sort column number (neg -> decending order)
NUM_INDE I Number of rows in table when it was last indexed.
INDEXnn I nn = 00 through 23. This is the first record
              number of an entry whose RA exceeds nn hours.
              Negative values indicate either no such entries or
              no index.

Table entries:

     Title Units Code Description
----------------- -------- ---- ------------------------------------
   RA(2000) Degree 1D Right Ascension (Equinox 2000)
   DEC(2000) Degree 1D Declination (Equinox 2000)
   PEAK INT Jy/beam 1E Peak Stokes I
   MAJOR AX Degree 1E Fitted major axis size
   MINOR AX Degree 1E Fitted minor axis size
   POSANGLE Degree 1E Fitted position angle of major axis
   Q CENTER Jy/beam 1E Interpolated Q value at position
                                  of I peak
   U CENTER Jy/beam 1E Interpolated U at I peak.
   P FLUX Jy 1E Integrated polarized (linear) flux.
   I RMS Jy/beam 1E RMS noise in Stokes I image
   POL RMS Jy/beam 1E RMS noise in Stokes Q,U image
   RES RMS Jy/beam 1E RMS Stokes I residual after fit.
   RES PEAK Jy/beam 1E Peak Stokes I residual.
   RES FLUX Jy 1E Integrated Stokes I residual.
   CENTER X pixels 1E "X" pixel in FIELD of center
   CENTER Y pixels 1E "Y" pixel in FIELD of center
   FIELD 8A Name of (4x4 deg) field
   JD PROCESSED Day 1I Julian date on which this entry
                                  was derived from an image.

2.3 Notes

The “code” column is element count + basic type code. basic type codes: D=Double precision, E=single precision, A=character, I=integer.

The value of “P FLUX” has been corrected for the amplitude bias; “Q CENTER” and “U CENTER” have not, nor have any of the Stokes I measures. The polarization measurments are only meaningful for an unresolved source as they do not measure variation across the source. RMS values are estimated from fitting a Gaussian to the central peak in the distribution of pixel values of the entire 4^∘ × 4^∘ image.

3 Fitting constraints

In the Gaussian fitting process neither the major nor minor axis sizes were allowed to be smaller than the restoring beam size. Also the sizes were not allowed to be larger than 9 cells (135”).

4 POSITION CALIBRATION 3

4 Position calibration

A comparison of NVSS positions with those in the USNO catalog of absolute radio source positions (Johnston et al. 1995, AJ, 110, 880.) indicate a bias of 0.”025 with an RMS scatter of 0.”45 in right ascension and a bias of -0.”113 and scatter of 0.”56 in declination. After this correction, the cataloged positions should be in the USNO frame. Note: this correction has not been made to the survey images, but is made in the NVSS catalog browser.

5 Confusion bias correction

The interaction of the NVSS dirty synthesized beam with the background of sources results in a systematic reduction of the source brightnesses in the image. To correct for this effect add 0.30 mJy − 0.25 mJy^2/peak to the peak Stokes I brightness where peak is the fitted flux density.

6 Parameter error estimates

This section discusses how to derive the uncertainties in the fitted model parameters (see Ball, 1975; Condon, 1997). This discussion will follow Condon, 1997 which includes the effects of correlated pixel noise. This development uses an effective signal–to–noise ratio (denoted by ρ) whose value depends on the parameter whose error is being estimated. The generalized signal–to–noise ratio is given by the heuristic expression:

ρ2 = θ_M θ_m 4θ^2_N [ 1 + ( θ_N θ_M )_2]_αM [ 1 + ( θ_N θ_m )_2]_αm A2 σ2 (1) where θ_N is the restoring beam FWHM (assumed to be the pixel noise correlation size scale), θ_M is the fitted Gaussian Major axis FWHM, and θ_m the FWHM of the minor axis. A is the peak value of the fitted Gaussian and σ is the RMS pixel noise in the map. α_M ≥ 1/2 and α_m ≥ 1/2 are free parameters determined by Monte Carlo tests.

In this discussion the value of ρ determined using the appropriate values of α_M and α_m in Eq. (1) will be used to determine the errors in parameters estimated from the Stokes’ I images and σ_{Q,U} , the RMS in the Q and U images, for errors in parameters estimates from the linear polarization images.

6.1 Peak Stokes I

The error in the peak Stokes I brightness (A) is given by: σ2_A ≈ 2A^2 ρ2 + (A ∗ 0.03)^2 + 0.0003^2 (2)

where ρ was determined from Eq. (1) using α_M = α_m = 3/2. The term 0.03 is a 3% calibration error and 0.0003 is the uncertainty in the bias correction. (There is a further discussion later giving corrections from unresolved sources.)

6.2 Major axis size

The error in the major axis size (θ_M ) is:

σ2_θM ≈

2θ^2_M ρ2 + (0.02 θ_N )^2 (3) which includes a 2% calibration error term and ρ was determined from Eq. (1) using α_M = 5/2, α_m = 1/2.

6.3 Minor axis size

The error in the minor axis size (θ_m) is:

σ2_θm ≈

2θ^2_m ρ2 + (0.02 θ_N )^2 (4)

which includes a 2% calibration error term and ρ was determined from Eq. (1) using α_M = 1/2, α_m = 5/2.

6.4 Position angle of major axis

The error in the position angle of the major axis in radians (ϕ) is: σ2_ϕ ≈ ( θ_M θ_m θ2_M − θ2_m )2 4 ρ2 (5) where ρ was determined from Eq. (1) using α_M = 1/2, α_m = 5/2.

6.5 Integrated Stokes I for an extended source

The integrated flux density of the component is related to the peak value by: I = A (θ_M θ_m θ2_N ) (6) The error in the integrated flux density when all parameters are fitted is given by: σ2_I ≈ I2 [σ2_A A2 + ( θ2_N θ_M θ_m )( σ2_θM θ2_M + σ2_θm θ2_m )] + (I ∗ 0.03)^2 + 0.0003^2 (7) Note that I is a flux density and A is a brightness in units of flux density per beam area.

6.6 Integrated Stokes I for an unresolved source

Errors in the angular size are anticorrelated with those in the peak value. If the angular size is known in advance and not fitted then the variance in A (σ^2_A) is halved. If the source is determined to be unresolved but its fit allowed the size to be a free parameter then the corrected, point source peak value is given by:

A^∗ = A ( θ_M θ_m θ^∗_M θ^∗_m )_{1/2} (8) where θ_M and θ_m are the fitted sizes and θ^∗_M and θ^∗_m are the assumed sizes. The assumed size for an unresolved source is that of the image restoring beam (θ_N ).

It is known a priori that most of the sources observed by the NVSS will be unresolved so if the fitted axis sizes are within the errors of the restoring beam size (45”) then the unresolved source case can be invoked with relative safety. The error in A^∗ is:

σ2_A� ≈

A^∗2 ρ2 + (A^∗ ∗ 0.03)^2 + 0.0003^2 (9)

For an unresolved source the integrated flux density equals A^∗ and its uncertainty equals σ_A� .

6.7 Integrated Stokes I for a partially resolved source

In the case of a source resolved on only the major axis define a pseudo peak value, A^†, from which the effects of the inappropriately fitted axis have been removed: A^† = A ( θ_m θ_N ) 1_2

The uncertainty in this value is given by:

σ2_Ay ≈ 3A^{†2} 2ρ^2

+ 0.0003^2

where ρ was determined from Eq. (1) using α_M = α_m = 3/2.

The best estimate of the integrated flux density of a partially resolved component is related to the peak value by: I = A^† ( θ_M θ_N ) (10)

The error in the integrated flux density in this case is given by: σ2_I ≈ I2 [ σ2_{Ay} A^†2 + ( θ_N θ_M )( σ2_θM θ2_M )] + (I ∗ 0.03)^2 (11)

Note that I is a flux density and A is a brightness in units of flux density per beam area.

6.8 Right Ascension

The positional uncertainties in the directions of the major and minor axes (σ^2_x, σ2_y) are given by: σ2_x ≈ 2 2θ^2_M 8ln2ρ^2 where ρ was determined from Eq. (1) using α_M = 5/2, α_m = 1/2. The first factor of 2 is required to adjust the errors into agreement with the more accurate FIRST positions.

The error in Right Ascension is σ2_{RA} = {sin2(ϕ)σ2_x + cos2(ϕ)σ2_y} + 0.”452 (12) which includes the 0.”45 calibration error.

6.9 Declination

The error in Declination is σ2_{dec} = {cos2(ϕ)σ2_x + sin2(ϕ)σ2_y} + 0.”562 (13) which includes the 0.”56 calibration error.

6.10 Integrated Polarized flux density

The estimated error in the integrated polarized flux density is given by: σ_{P ol flux} = √ 2 σ_{Q,U} ( θ_M θ_m θ2_N ) . (14) .

7 POLARIZATION AMPLITUDE BIAS CORRECTION 6

6.11 Central polarization angle

The uncertainty in the polarization angle at the position of the peak in Stokes I in degrees is given by:

σ_{P ol P A} = 28.6479 σ_{Q,U} P ol Amp (15)

where P ol Amp is the linearly polarized amplitude ( √ peak2_Q + peak2_U ) at the position of the peak in Stokes I after correction for the amplitude bias.

7 Polarization amplitude bias correction

Since the distribution of expected pixel values in polarization amplitude ( √ Q2 + U 2) is not peaked at zero there is a bias in this measure which must be corrected at low signal–to–noise ratio. This correction can be made using the Fortran subroutine which follows. The comments describe the general method.

      SUBROUTINE PDBIAS (P, RMS)
C-----------------------------------------------------------------------
C Estimates the polarization bias in a polarization amplitude, P,
C measured in the presence of Q and U RMS noise, RMS. Returns the
C corrected value.
C The bias correction is such that the average bias is removed;
C thus the average in the absence of a signal is zero. Does table
C look of values calculated by J. Condon in the range of P/RMS of 1.253
C (the no signal limit) and 4 (the high SNR regime). Does second
C order Lagrange interpolation. At lower values of P/RMS the bias is
C a constant 1.253*RMS. Above a signal-to-noise ratio of 4, use the
C formula:
C normalized bias = 1 / (2 * s) + 1 / (8 * s**3),
C where s is the true normalized flux density, iterating once to
C estimate s from the normalized map flux density. "Normalized" means
C divided by the rms noise in the q and u maps.
C Inputs:
C P R On input, P is the observed total polarized intensity
C RMS R The standard deviation of the (assumed equal)
C Gaussian distributions of the Stokes Q or U maps.
C Output:
C P R On output, P is the estimated intrinsic total
C polarized intensity.
C-----------------------------------------------------------------------
      REAL P, RMS
C
      INTEGER I, INDEX, I1, I2, I3
      REAL TABLE(2,40), PNORM, BIAS, D1, D2, D3, WT1, WT2, WT3,
     * SUM, SUMWT
C (map_flux,map_bias) pairs
      DATA TABLE /
     * 1.253,1.253, 1.256,1.156, 1.266,1.066, 1.281,0.9814,
     * 1.303,0.9030, 1.330,0.8304, 1.364,0.7636, 1.402,0.7023,
     * 1.446,0.6462, 1.495,0.5951, 1.549,0.5486, 1.606,0.5064,
     * 1.668,0.4683, 1.734,0.4339, 1.803,0.4028, 1.875,0.3749,
     * 1.950,0.3498, 2.027,0.3273, 2.107,0.3070, 2.189,0.2888,
     * 2.272,0.2724, 2.358,0.2576, 2.444,0.2442, 2.532,0.2321,
     * 2.621,0.2212, 2.711,0.2112, 2.802,0.2021, 2.894,0.1938,
     * 2.986,0.1861, 3.079,0.1791, 3.173,0.1726, 3.267,0.1666,
     * 3.361,0.1610, 3.456,0.1557, 3.551,0.1509, 3.646,0.1463,
     * 3.742,0.1420, 3.838,0.1380, 3.934,0.1342, 4.031,0.1306/

C-----------------------------------------------------------------------
C Check RMS
      IF (RMS.LE.0.0) GO TO 999
      PNORM = P / RMS
C Which regime?
      IF (PNORM.LE.TABLE(1,1)) THEN
C Low (no) SNR case
         P = P - TABLE(2,1) * RMS
      ELSE IF (PNORM.GE.TABLE(1,40)) THEN
C High SNR
         BIAS = 1.0 / (2.0 * PNORM) + 1.0 / (8.0 * PNORM**3)
         PNORM = PNORM - BIAS
         BIAS = 1.0 / (2.0 * PNORM) + 1.0 / (8.0 * PNORM**3)
C Correct for bias

8 DECONVOLUTION 7

         P = P - BIAS * RMS
      ELSE
C Middle, interpolate in table
         INDEX = 1
         DO 20 I = 3,39
            IF (PNORM.LT.TABLE(1,I)) GO TO 30
            INDEX = I
 20 CONTINUE
C Lagrange interpolation
 30 I1 = INDEX - 1
         I2 = INDEX
         I3 = INDEX + 1
         D1 = (TABLE(1,I1) - TABLE(1,I2)) * (TABLE(1,I1) - TABLE(1,I3))
         D2 = (TABLE(1,I2) - TABLE(1,I1)) * (TABLE(1,I2) - TABLE(1,I3))
         D3 = (TABLE(1,I3) - TABLE(1,I1)) * (TABLE(1,I3) - TABLE(1,I2))
         WT1 = (PNORM - TABLE(1,I2)) * (PNORM - TABLE(1,I3)) / D1
         WT2 = (PNORM - TABLE(1,I1)) * (PNORM - TABLE(1,I3)) / D2
         WT3 = (PNORM - TABLE(1,I1)) * (PNORM - TABLE(1,I2)) / D3
         SUM = TABLE(2,I1) * WT1 + TABLE(2,I2) * WT2 + TABLE(2,I3) * WT3
         SUMWT = WT1 + WT2 + WT3
         IF (SUMWT.GT.0.0) THEN
            BIAS = SUM / SUMWT
         ELSE
C Shouldn’t ever get here but do
C something reasonable.
            BIAS = TABLE(2,I2)
            END IF
C Correct for bias
         P = P - BIAS * RMS
         END IF
C
 999 RETURN
      END

8 Deconvolution

The values given in the table are fitted Gaussians and as such represent the convolution of the sky with the restoring beam. The following Fortran routine represents a method of determining the appropriate Gaussian axis sizes of the source structure with error estimates for a circular restoring beam. In this case, the source position angle is that of the fitted Gaussian with the same uncertainty.

      SUBROUTINE DECONV (BWID, FMAJ, UFMAJ, FMIN, UFMIN, SMAJ, UPSMAJ,
     * UMSMAJ, SMIN, UPSMIN, UMSMIN)
C-----------------------------------------------------------------------
C This subroutine deconvolves a fitted elliptical gaussian into an
C elliptical gaussian source and a CIRCULAR gaussian beam, and it
C calculates the errors in the deconvolved sizes. The arguments are
C dimensionless angles, so any single unit (e.g., deg or arcsec) may
C be used for all of them. The fitted major and minor axes are
C compared with the beamwidth to see if the source is well resolved,
C that is, the fitted axis is some number of sigma bigger than the
C beam (e.g., SIGMIN = 2.33 => 98% confidence that the source is
C resolved). If a source axis is well resolved, both its +rms error
C and -rms are calculated. The +/-rms error is the difference between
C the deconvolved size and the deconvolved size corresponding to the
C fitted size +/- its rms error. Unless the source is just barely
C resolved, the +rms and -rms errors are nearly equal and can be
C averaged. If either source axis is not well resolved, then that axis
C (either SMAJ or SMIN) is set to an upper limit whose significance is
C determined by SIGMAX (e.g., SIGMAX = 2.33 => 98% confidence upper
C limit). If the fitted size is significantly smaller than the beam
C (an "impossble", or at least "improbable", source), the source axis
C size is set to -1. The errors associated with unresolved source axes
C (e.g., UMSMAJ, UMSMIN, UPSMAJ, UPSMIN) are set to -1. to indicate
C that the quoted sizes are upper limits.
C Version: 95/05/13 J. J. Condon
C 97/12/10 Corrected error in calculation of
C rms uncertainties in deconvolved sizes J. J. Condon
C Inputs:
C BWID R Beam circular gaussian FWHM
C FMAJ R Fitted elliptical gaussian FWHM major axis
C FMIN R Fitted elliptical gaussian FWHM minor axis
C UFMAJ R RMS error in FMAJ
C UFMIN R RMS error in FMIN
C Outputs:

8 DECONVOLUTION 8

C SMAJ R Source elliptical gaussian FWHM major axis
C SMIN R Source elliptical gaussian FWHM minor axis
C UMSMAJ R -RMS error in SMAJ
C UMSMIN R -RMS error in SMIN
C UPSMAJ R +RMS error in SMAJ
C UPSMIN R +RMS error in SMIN
C Control parameters:
C SIGMAX R Number of UFMAJ, UFMIN to add to set source size limit
C SIGMIN R Number of UFMAJ, UFMIN to decide if source well
C resolved
C (for both sigmax and sigmin, 2.33 => 98% confidence ---
C see Bevington, P. R., Data Reduction and Error Analysis for the
C Physical Sciences, Table C-2)
C-----------------------------------------------------------------------
      REAL BWID, FMAJ, FMIN, UFMAJ, UFMIN, SMAJ, SMIN, UMSMAJ,
     * UMSMIN, UPSMAJ, UPSMIN
C
      REAL SIGMAX, SIGMIN
      REAL TEST, RADARG
C Confidence-level parameters
C sigmax, sigmin: 1.28 => 80%,
C 1.65 => 90%, 2.33 => 98%
C Both must be > 1.00 for
C equations to be valid.

      PARAMETER (SIGMAX = 2.33)
      PARAMETER (SIGMIN = 2.33)
C-----------------------------------------------------------------------
C Calculate source FWHM major axis
      SMAJ = FMAJ * FMAJ - BWID * BWID
      IF (SMAJ .GT. 0.) THEN
         SMAJ = SQRT (SMAJ)
      ELSE
         SMAJ = 0.
      END IF
C Calculate source FWHM minor
C axis
      SMIN = FMIN * FMIN - BWID * BWID
      IF (SMIN .GT. 0.) THEN
         SMIN = SQRT (SMIN)
      ELSE
         SMIN = 0.
      END IF
C Test for significant
C major-axis resolution
      TEST = (FMAJ - UFMAJ * SIGMIN)
      TEST = TEST * TEST - BWID * BWID
      IF (TEST .GT. 0.) THEN
C Source major axis is well
C resolved. Calculate source
C major axis - rms error.
         RADARG = (FMAJ - UFMAJ) * (FMAJ - UFMAJ) - BWID * BWID
         UMSMAJ = SMAJ - SQRT (RADARG)
C Calculate source major axis
C +rms error.
         RADARG = (FMAJ + UFMAJ) * (FMAJ + UFMAJ) - BWID * BWID
         UPSMAJ = SQRT(RADARG) - SMAJ
      ELSE
C Source major axis is not
C significantly resolved.
C Calculate source major axis
C upper limit.
         SMAJ = FMAJ + SIGMAX * UFMAJ
         IF (SMAJ .GT. BWID) THEN
            RADARG = SMAJ * SMAJ - BWID * BWID
            SMAJ = SQRT (RADARG)
         ELSE
            SMAJ = -1.
            END IF
C Set source uncertainties to
C -1
         UPSMAJ = -1.
         UMSMAJ = -1.
         END IF
C Test for significant
C minor-axis resolution
      TEST = (FMIN - UFMIN * SIGMIN)
      TEST = TEST * TEST - BWID * BWID
      IF (TEST .GT. 0.) THEN
C Source minor axis is well
C resolved. Calculate source

REFERENCES 9

C minor axis -rms error.
         RADARG = (FMIN - UFMIN) * (FMIN - UFMIN) - BWID * BWID
         UMSMIN = SMIN - SQRT (RADARG)
C Calculate source minor axis
C +rms error.
         RADARG = (FMIN + UFMIN) * (FMIN + UFMIN) - BWID * BWID
         UPSMIN = SQRT(RADARG) - SMIN
      ELSE
C Source minor axis is not
C significantly resolved.
C Calculate source minor axis
C upper limit
         SMIN = FMIN + SIGMAX * UFMIN
         IF (SMIN .GT. BWID) THEN
            RADARG = SMIN * SMIN - BWID * BWID
            SMIN = SQRT(RADARG)
         ELSE
            SMIN = -1.
            END IF
C Set source uncertainties to
C -1
         UPSMIN = -1.
         UMSMIN = -1.
         END IF
C
 999 RETURN
      END

References

[1] Ball, J. 1975, Methods in Computational Physics, 14, 177.

[2] Condon, J. J., “Errors in Elliptical Gaussian Fits” 1997, P.A.S.P. 109, 166–

172.

[3] Condon, J. J., et al. 1998 in preparation.

[4] Cotton, W. D., Tody, D., Pence, W. D. 1995, A&A Supp., 112, 159.